

**EECE 311  
Fall 2011**

**Lab Report #4  
Modeling and Analyzing AC Circuits (RC and RL) with SPICE  
10/4/2011**

Signature	Printed Name	Date
	<b>Jeremiah Mahler</b>	<b>Oct 11, 2011</b>

### **1 Objective**

The objective of this laboratory exercise is to learn and gain experience analyzing and simulating the behavior of first order RC and RL circuits under transient conditions using a SPICE simulator.

### **2 Equipment**

To perform the circuit simulation the Ngspice[1] SPICE[2] simulator was used. Other SPICE simulators such as Pspice[3] and Orcad[4] should work as well.

### **3 Procedure**

Two circuits will be analyzed; one is an RC with a capacitor and the other is an RL with a inductor. For each of these circuits this experiment consists of two main steps.

1. Build a SPICE definition of the circuit.
2. Run the simulation and record the output.

In general a SPICE simulation can be run using Ngspice with the command

```
ngspice -b your_file.cir
```

where `your_file.cir` replaced by the name of your file containing the SPICE definition. To save the output to a file a redirect can be used as in:

```
ngspice -b your_file.cir > your_file.out
```

### 3.1 RC Circuit

The RC circuit to be analyzed is shown in Figure 1. The SPICE definition is shown in Figure 2.




Figure 1: RC Circuit definition. Nodes denoted by numbers with 0 as common.  $V1$  is a voltage source,  $I1$  is a current source. For the current source  $I1$ ,  $u(t)$  is the unit step function which has a value of 0 until  $t > 0$ . This causes the current source to change from 0 amps to 6 mA at  $t = 0$ .

```

Voltage vs Time, RC circuit
* See below for instructions for how to run this file.

* Voltage and current source orientations.
* Vx + -
* Ix - + (out +)

V1 1 0 DC 3V
R1 1 2 2k

* PULSE(min max time_delay rise_time fall_time pulse_width [period])
*I1 2 0 DC PULSE(0 3mA 0 1NS 1NS 100US 200US)
* PWL(t0 v0 t1 v1 t2 v2 ...)
I1 2 0 DC PWL(0 0 0 6mA)

R2 2 3 1k
R3 3 0 6k
C1 3 0 1nF

* Uncomment to choose the mode your are running:
*--> Non-interactive mode <-----
* ngspice -b rc_circuit-01.cir

** .TRAN step end_time
*.TRAN 1uS 24uS
*.PLOT TRAN V(1) V(2) V(3)
**.PROBE

*--> Interactive mode <-----
* ngspice rc_circuit-01.cir

.CONTROL

*      step  t_max
TRAN 1uS    12us

* gui (X11) plot
*PLOT V(1) V(2) V(3)

GNUPLOT plot-01 V(1) V(2) V(3)
* gnuplot -persist plot-01.plt
* (outputs to plot-01.eps)
QUIT

.ENDC
*-----
.END

```

Figure 2: SPICE definition of RC circuit in Figure 1.

### 3.2 RL circuit

The RL circuit to be analyzed is shown in Figure 3. The SPICE definition is shown in Figure 4.




Figure 3: RL Circuit definition. Nodes denoted by numbers with 0 as common. V1 is an alternating current voltage source, VB is a direct current voltage source, L1 is an inductor.

```

RL circuit with alternating source, 169 sin(377 t)
* See below for instructions for how to run this file.

* Voltage and current source orientations.
* Vx + -
* Ix - + (out +)

* 169 sin(377 t)
* 377 / (2*pi) [rad] = 60 [Hz]
* SIN(V0 VA FREQ TD THETA)
V1 1 0 SIN(0 169 60)
R1 1 2 10
L1 2 3 25mH
VB 3 0 80

* Uncomment to choose the mode your are running:
*--> Non-interactive mode <-----
* ngspice -b rc_circuit-01.cir

* .TRAN step end_time
*.TRAN 200e-6 30ms
*.TRAN 200e-6 40ms
*.PLOT TRAN V(1) V(2) V(2,3)

*--> Interactive mode <-----
** ngspice rc_circuit-01.cir
*
.CONTROL
*
** step t_max
*TRAN 200e-6 30ms
TRAN 200e-6 40ms
*
** gui (X11) plot
**PLOT V(1) V(2) V(2,3)
*
GNUPLOT rlplot-01 V(1) V(2) V(2,3)
** gnuplot -persist rlplot-01.plt
** (outputs to rlplot-01.eps)
QUIT
*
.ENDC
*-----
.END

```

Figure 4: SPICE definition of RL circuit in Figure 3.

## 4 Results

### 4.1 RC circuit

The output of the SPICE simulation of the RC circuit (Figure 1) is shown in Figure 5.




Figure 5: Plot of SPICE simulation (Figure 2) for the RC circuit (Figure 1). The change of the current source from zero current to 6 mA results in charging of the capacitor. The voltage is negative as a result of the sign orientation used in the circuit.

## 4.2 RL circuit

The output of the SPICE simulation of the RL circuit (Figure 3) is shown in Figure 6.




Figure 6: Plot of SPICE simulation (Figure 4) for the RL circuit (Figure 3). Notice that during the first cycle it has not yet reached the steady state because the amplitude is lower than during subsequent cycles. The voltage across the inductor ( $v(2,3)$ ) is offset compared to the alternating voltage source ( $v_1$ ) and is shifted due to the DC voltage source ( $v_B$ ).

## 5 Correlation with theory

To correlate the simulation with theory manual calculations are performed.

### 5.1 RC circuit

The first stage of analysis is to determine the steady state conditions at  $t = 0^-$ , before the circuit before switch point of the unit step function ( $u(t)$ ). Referring to Figure 1, it can be seen that when the current source is flowing 0 amps the open circuit voltage across the capacitor can be calculated as a voltage divider.

$$\begin{aligned} v_3(0^-) &= 3 \cdot \frac{6k}{6k + 2k + 1k} \\ &= 2.0 \text{ [volts]} \end{aligned}$$

The second stage is to simplify the circuit in to an equivalent circuit with only one resistor, one capacitor and one voltage source. Figure 7 shows the result of this simplification.




Figure 7: Simplified form of RC circuit from Figure 1

Using the simplified circuit (Figure 7) along with Kirchoff's voltage law and the equation for a capacitor the behavior of the circuit is given by Equation 1.

$$6 + RC \frac{dv}{dt} + v = 0 \quad (1)$$



And solving this first order differential equation:

$$\begin{aligned}
-RC \frac{dv}{dt} &= 6 + v \\
-RC \frac{dv}{v + 6} &= dt \\
\int_{v(0)}^{v(t)} \frac{dv}{v + 6} &= -\frac{1}{RC} \int_0^t dt \\
-\frac{t}{RC} &= [\ln(v(t) + 6) - \ln(v(0) + 6)] \\
-\frac{t}{RC} &= \ln \left( \frac{v(t) + 6}{v(0) + 6} \right) \\
e^{-t/(RC)} &= \frac{v(t) + 6}{v(0) + 6} \\
v(t) + 6 &= (v(0) + 6)e^{-t/(RC)} \\
v(t) &= -6 + (v(0) + 6)e^{-t/(RC)}
\end{aligned}$$

And substituting the actual parameters along with the initial voltage:

$$\begin{aligned}
v(t) &= -6 + (2.0 + 6)e^{-t \cdot 500 \times 10^3} \quad [\text{volts}] \\
v(t) &= -6 + 8e^{-t \cdot 500 \times 10^3} \quad [\text{volts}]
\end{aligned} \tag{2}$$

Next, values are substituted in to Equation 2 and compared to the plot (Figure 5). As  $t$  goes to  $\infty$  the voltage across the capacitor goes to  $-6$  volts. At  $t = 0$  the voltage across the capacitor is 2.0 volts. At an arbitrary point,  $t = 2RC$ , the voltage across the capacitor is -4.92 volts. All these values agree with the plot from the SPICE simulation.

## 5.2 RL circuit

There are two possible ways to analyze the RL circuit with an alternating voltage source. The first is by using differential equations. This is the most complex method but it completely describes the behavior including transient conditions. The second is to analyze only during steady state conditions [6, Pg. 330]. This method uses the concept of impedance and is much simpler but excludes the transient conditions. To simplify this analysis the second method will be used.

First the impedances of each passive circuit element is found.

$$\begin{aligned}
Z_{R1} &= 10 \\
Z_{L1} &= j\omega L \\
&= j9.425 \\
Z_{eq} &= Z_{R1} + Z_{L1} \\
&= 10 + j9.425
\end{aligned}$$

Since the effect of the DC voltage source ( $v_B$ ) only serves to shift the voltage it will not be included in the intermediate calculations. It will be accounted for at the end.

The voltage source must be converted to the complex plane.

$$\begin{aligned}
V_1 &= 169 \sin(377t) \\
&= 169 \cos(377t - 90) \\
&= 169 \cos(377t - \pi/2) \\
&= 169 \angle -90^\circ \\
&= 0 - j169
\end{aligned}$$

The current is found in the complex plane by using the voltage ( $V_1$ ) and the equivalent impedance ( $Z_{eq}$ ).

$$\begin{aligned}
I &= \frac{V_1}{Z_{eq}} \\
&= \frac{0 - j169}{10 + j9.425} \\
&= -8.435 - j8.949
\end{aligned}$$

The voltage across the inductor is found using the current ( $I$ ) and the impedance ( $Z_L$ ).

$$\begin{aligned}
V_L &= IZ_L \\
&= (-8.435 - j8.949)(j9.425) \\
&= 84.35 - j7.950 \\
&= 115.913 \angle -43.30^\circ \\
&= 115.913 \cos(377t - 43.30^\circ) \\
&= 115.913 \cos(377t - 0.756)
\end{aligned}$$

And to find the voltage at node 2 the DC voltage source is added back in.

$$\begin{aligned} V_2 &= V_L + V_B \\ V_2 &= 115.913 \cos(377t - 0.756) + 80 \end{aligned} \tag{3}$$

Equation 3 describes the voltage at node 2 ( $V_2$ ) over time. Substituting for various arbitrary values: at  $t = 0.01875$   $V_2 = 115.86$ , at  $t = 0.01$   $V_2 = -114.971$ . These values agree approximately with the plot of SPICE output (Figure 6).

## 6 Conclusion

This experiment was a success in analyzing the behavior of first order RC and RL circuits under transient conditions. The calculated values approximately matched the plotted values from the SPICE output and agreed with the theory.

## 7 References

- [1] Ngspice, “Ngspice, spice circuit simulator.” <http://ngspice.sourceforge.net>, 2011.
- [2] Wikipedia, “Spice — Wikipedia, the free encyclopedia.” <http://en.wikipedia.org/wiki/SPICE>, 2011. [Online; accessed 19-September-2011].
- [3] Wikipedia, “Pspice — Wikipedia, the free encyclopedia.” <http://en.wikipedia.org/wiki/Pspice>, 2011. [Online; accessed 8-October-2011].
- [4] “Orcad, electronic design automation software.” <http://www.cadence.com/orcad/>, 2011. Cadence Design Systems.
- [5] T. Williams and C. Kelley, “Gnuplot.” <http://www.gnuplot.info>, 2011.
- [6] J. Nilsson and S. Riedel, Electric circuits. Pearson/Prentice Hall, 2008.